

Postestimation commands

The following postestimation commands are available after `xhtaylor`:

Command	Description
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
<code>forecast</code>	dynamic forecasts and simulations
<code>hausman</code>	Hausman’s specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of parameters
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of parameters
<code>predict</code>	predictions and their SEs, residuals, etc.
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

predict

Description for predict

predict creates a new variable containing predictions such as fitted values, standard errors, combined residuals, predictions, random-error components, and idiosyncratic error components.

Menu for predict

Statistics > Postestimation

Syntax for predict

```
predict [type] newvar [if] [in] [ , statistic]
```

statistic	Description
Main	
xb	$\mathbf{X}_{it}\widehat{\boldsymbol{\beta}} + \mathbf{Z}_i\widehat{\boldsymbol{\delta}}$, fitted values; the default
stdp	standard error of the fitted values
ue	$\hat{\mu}_i + \hat{\epsilon}_{it}$, the combined residual
* xbu	$\mathbf{X}_{it}\widehat{\boldsymbol{\beta}} + \mathbf{Z}_i\widehat{\boldsymbol{\delta}} + \hat{\mu}_i$, prediction including effect
* u	$\hat{\mu}_i$, the random-error component
* e	$\hat{\epsilon}_{it}$, prediction of the idiosyncratic error component

Unstarred statistics are available both in and out of sample; type predict ... if e(sample) ... if wanted only for the estimation sample. Starred statistics are calculated only for the estimation sample, even when if e(sample) is not specified.

Options for predict

Main

- xb, the default, calculates the linear prediction, that is, $\mathbf{X}_{it}\widehat{\boldsymbol{\beta}} + \mathbf{Z}_i\widehat{\boldsymbol{\delta}}$.
- stdp calculates the standard error of the linear prediction.
- ue calculates the prediction of $\hat{\mu}_i + \hat{\epsilon}_{it}$.
- xbu calculates the prediction of $\mathbf{X}_{it}\widehat{\boldsymbol{\beta}} + \mathbf{Z}_i\widehat{\boldsymbol{\delta}} + \hat{\mu}_i$, the prediction including the random effect.
- u calculates the prediction of $\hat{\mu}_i$, the estimated random effect.
- e calculates the prediction of $\hat{\epsilon}_{it}$.

margins

Description for margins

margins estimates margins of response for fitted values.

Menu for margins

Statistics > Postestimation

Syntax for margins

```
margins [marginlist] [ , options ]
margins [marginlist] , predict(statistic ...) [options]
```

<i>statistic</i>	Description
xb	$\mathbf{X}_{it}\widehat{\boldsymbol{\beta}} + \mathbf{Z}_i\widehat{\boldsymbol{\delta}}$, fitted values; the default
stdp	not allowed with margins
ue	not allowed with margins
xbu	not allowed with margins
u	not allowed with margins
e	not allowed with margins

Statistics not allowed with margins are functions of stochastic quantities other than $e(b)$.

For the full syntax, see [R] [margins](#).

Remarks and examples

► Example 1

Continuing with [example 1](#) of [XT] [xhtaylor](#), we use `hausman` to test whether we should use the Hausman–Taylor estimator instead of the fixed-effects estimator. We follow the empirical illustration in [Baltagi \(2013, sec. 7.5\)](#), but we fit the model without including the `exp2` and `wks` variables.

We first fit the model with `xhtaylor` and then with `xtreg, fe`:

```
. use https://www.stata-press.com/data/r19/psidextract
. xhtaylor lwage occ south smsa ind exp ms union fem blk ed,
> endog(exp ms union ed)
  (output omitted)
. estimates store eq_ht
. xtreg lwage occ south smsa ind exp ms union fem blk ed, fe
  (output omitted)
. estimates store eq_fe
```

We can now use `hausman` to compare the two estimators, but we need to specify the `df()` to indicate the degrees of freedom for the χ^2 statistic, which would be determined by the overidentifying restrictions in the Hausman–Taylor estimation. In this case, there are three degrees of freedom because there are four time-varying exogenous variables (`occ`, `south`, `smsa`, `ind`) that can be used as instruments for only one time-invariant endogenous variable (`ed`).

```
. hausman eq_fe eq_ht, df(3)
```

	Coefficients		(b-B) Difference	sqrt(diag(V_b-V_B)) Std. err.
	(b) eq_fe	(B) eq_ht		
occ	-.0239323	-.0231694	-.0007629	.0002395
south	-.0037282	.0062699	-.0099982	.0124188
smsa	-.0436251	-.0433518	-.0002733	.0042296
ind	.021184	.0156376	.0055465	.0025159
exp	.0965738	.0964748	.0000991	.000063
ms	-.0299908	-.0300703	.0000795	.000321
union	.0349156	.0348494	.0000662	.0006336

b = Consistent under H0 and Ha; obtained from **xtreg**.

B = Inconsistent under Ha, efficient under H0; obtained from **xhtaylor**.

Test of H0: Difference in coefficients not systematic

chi2(3) = (b-B)'[(V_b-V_B)⁻¹](b-B)
= 5.22

Prob > chi2 = 0.1567

(V_b-V_B is not positive definite)

The *p*-value for the test provides evidence favoring the null hypothesis; therefore, in this case, the Hausman–Taylor estimation is adequate.

Notice that the variance–covariance matrix for the difference (b–B) is not positive definite. As [Greene \(2012, 237\)](#) points out, this kind of result is due to finite-sample conditions. He also states that Hausman considers it preferable to take the test statistic as zero and, therefore, not to reject the null hypothesis.



► Example 2

We now want to determine whether the Amemiya–MaCurdy estimator produces significant efficiency gains with respect to the Hausman–Taylor estimator. We refit the two models, and we use the Hausman test again:

```
. use https://www.stata-press.com/data/r19/psidextract
. xhtaylor lwage occ south smsa ind exp ms union fem blk ed,
> endog(exp ms union ed)
(output omitted)
. estimates store eq_ht
. xhtaylor lwage occ south smsa ind exp ms union fem blk ed,
> endog(exp ms union ed) amacurdy
(output omitted)
. estimates store eq_am
```

```
. hausman eq_ht eq_am
```

	Coefficients			
	(b) eq_ht	(B) eq_am	(b-B) Difference	$\sqrt{\text{diag}(V_b - V_B)}$ Std. err.
occ	-.0231694	-.023354	.0001846	.0006485
south	.0062699	.0060857	.0001842	.0010641
smsa	-.0433518	-.0434638	.0001121	.0006297
ind	.0156376	.0156602	-.0000226	.000492
exp	.0964748	.0962147	.00026	.0000694
ms	-.0300703	-.0303139	.0002436	.0006735
union	.0348494	.0345742	.0002752	.0006471
fem	-.1277756	-.1287857	.0010101	.0036717
blk	-.2911574	-.291645	.0004876	.0082831
ed	.1390257	.1380699	.0009558	.005436

b = Consistent under H0 and Ha; obtained from **xhtaylor**.

B = Inconsistent under Ha, efficient under H0; obtained from **xhtaylor**.

Test of H0: Difference in coefficients not systematic

```
chi2(10) = (b-B)'[(V_b-V_B)^(-1)](b-B)
          = 14.42
```

```
Prob > chi2 = 0.1548
```

The result indicates that we should use the more efficient estimation produced by the Amemiya–MacCurdy estimator.

◀

References

Baltagi, B. H. 2013. *Econometric Analysis of Panel Data*. 5th ed. Chichester, UK: Wiley.

Greene, W. H. 2012. *Econometric Analysis*. 7th ed. Upper Saddle River, NJ: Prentice Hall.

Also see

[XT] **xhtaylor** — Hausman–Taylor estimator for error-components model

[U] **20 Estimation and postestimation commands**

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).